

## NAG Library Routine Document

### F08JAF (DSTEV)

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

#### 1 Purpose

F08JAF (DSTEV) computes all the eigenvalues and, optionally, all the eigenvectors of a real  $n$  by  $n$  symmetric tridiagonal matrix  $A$ .

#### 2 Specification

```
SUBROUTINE F08JAF (JOBZ, N, D, E, Z, LDZ, WORK, INFO)
  INTEGER          N, LDZ, INFO
  REAL (KIND=nag_wp) D(*), E(*), Z(LDZ,*), WORK(*)
  CHARACTER(1)    JOBZ
```

The routine may be called by its LAPACK name *dste*.

#### 3 Description

F08JAF (DSTEV) computes all the eigenvalues and, optionally, all the eigenvectors of  $A$  using a combination of the  $QR$  and  $QL$  algorithms, with an implicit shift.

#### 4 References

Anderson E, Bai Z, Bischof C, Blackford S, Demmel J, Dongarra J J, Du Croz J J, Greenbaum A, Hammarling S, McKenney A and Sorensen D (1999) *LAPACK Users' Guide* (3rd Edition) SIAM, Philadelphia <http://www.netlib.org/lapack/lug>

Golub G H and Van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

#### 5 Arguments

- 1: JOBZ – CHARACTER(1) *Input*  
*On entry:* indicates whether eigenvectors are computed.  
 JOBZ = 'N'  
     Only eigenvalues are computed.  
 JOBZ = 'V'  
     Eigenvalues and eigenvectors are computed.  
*Constraint:* JOBZ = 'N' or 'V'.
- 2: N – INTEGER *Input*  
*On entry:*  $n$ , the order of the matrix.  
*Constraint:*  $N \geq 0$ .
- 3: D(\*) – REAL (KIND=nag\_wp) array *Input/Output*  
**Note:** the dimension of the array D must be at least  $\max(1, N)$ .  
*On entry:* the  $n$  diagonal elements of the tridiagonal matrix  $A$ .  
*On exit:* if INFO = 0, the eigenvalues in ascending order.

- 4:  $E(*)$  – REAL (KIND=nag\_wp) array *Input/Output*  
**Note:** the dimension of the array  $E$  must be at least  $\max(1, N - 1)$ .  
*On entry:* the  $(n - 1)$  subdiagonal elements of the tridiagonal matrix  $A$ .  
*On exit:* the contents of  $E$  are destroyed.
- 5:  $Z(\text{LDZ}, *)$  – REAL (KIND=nag\_wp) array *Output*  
**Note:** the second dimension of the array  $Z$  must be at least  $\max(1, N)$  if  $\text{JOBZ} = 'V'$ , and at least 1 otherwise.  
*On exit:* if  $\text{JOBZ} = 'V'$ , then if  $\text{INFO} = 0$ ,  $Z$  contains the orthonormal eigenvectors of the matrix  $A$ , with the  $i$ th column of  $Z$  holding the eigenvector associated with  $D(i)$ .  
 If  $\text{JOBZ} = 'N'$ ,  $Z$  is not referenced.
- 6: LDZ – INTEGER *Input*  
*On entry:* the first dimension of the array  $Z$  as declared in the (sub)program from which F08JAF (DSTEV) is called.  
*Constraints:*  
     if  $\text{JOBZ} = 'V'$ ,  $\text{LDZ} \geq \max(1, N)$ ;  
     otherwise  $\text{LDZ} \geq 1$ .
- 7: WORK(\*) – REAL (KIND=nag\_wp) array *Workspace*  
**Note:** the dimension of the array WORK must be at least  $\max(1, 2 \times N - 2)$ .  
*On exit:* if  $\text{JOBZ} = 'N'$ , WORK is not referenced.
- 8: INFO – INTEGER *Output*  
*On exit:*  $\text{INFO} = 0$  unless the routine detects an error (see Section 6).

## 6 Error Indicators and Warnings

INFO < 0

If  $\text{INFO} = -i$ , argument  $i$  had an illegal value. An explanatory message is output, and execution of the program is terminated.

INFO > 0

If  $\text{INFO} = i$ , the algorithm failed to converge;  $i$  off-diagonal elements of  $E$  did not converge to zero.

## 7 Accuracy

The computed eigenvalues and eigenvectors are exact for a nearby matrix  $(A + E)$ , where

$$\|E\|_2 = O(\epsilon)\|A\|_2,$$

and  $\epsilon$  is the *machine precision*. See Section 4.7 of Anderson *et al.* (1999) for further details.

## 8 Parallelism and Performance

F08JAF (DSTEV) is threaded by NAG for parallel execution in multithreaded implementations of the NAG Library.

F08JAF (DSTEV) makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor library used by this implementation. Consult the documentation for the vendor library for further information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

## 9 Further Comments

The total number of floating-point operations is proportional to  $n^2$  if `JOBZ = 'N'` and is proportional to  $n^3$  if `JOBZ = 'V'`.

## 10 Example

This example finds all the eigenvalues and eigenvectors of the symmetric tridiagonal matrix

$$A = \begin{pmatrix} 1 & 1 & 0 & 0 \\ 1 & 4 & 2 & 0 \\ 0 & 2 & 9 & 3 \\ 0 & 0 & 3 & 16 \end{pmatrix},$$

together with approximate error bounds for the computed eigenvalues and eigenvectors.

### 10.1 Program Text

```

Program f08jafe

!      F08JAF Example Program Text

!      Mark 26 Release. NAG Copyright 2016.

!      .. Use Statements ..
      Use nag_library, Only: ddisna, dstev, nag_wp, x02ajf, x04caf
!      .. Implicit None Statement ..
      Implicit None
!      .. Parameters ..
      Integer, Parameter          :: nin = 5, nout = 6
!      .. Local Scalars ..
      Real (Kind=nag_wp)          :: eerrbd, eps
      Integer                     :: i, ifail, info, ldz, n
!      .. Local Arrays ..
      Real (Kind=nag_wp), Allocatable :: d(:), e(:), rcondz(:), work(:),      &
                                         z(:,,:), zerrbd(:)
!      .. Intrinsic Procedures ..
      Intrinsic                   :: abs, max
!      .. Executable Statements ..
      Write (nout,*) 'F08JAF Example Program Results'
      Write (nout,*)
!      Skip heading in data file
      Read (nin,*)
      Read (nin,*) n
      ldz = n
      Allocate (d(n),e(n-1),rcondz(n),work(2*n-2),z(ldz,n),zerrbd(n))

!      Read the diagonal and off-diagonal elements of the matrix A
!      from data file

      Read (nin,*) d(1:n)
      Read (nin,*) e(1:n-1)

!      Solve the symmetric tridiagonal eigenvalue problem
!      The NAG name equivalent of dstev is f08jaf
      Call dstev('Vectors',n,d,e,z,ldz,work,info)

      If (info==0) Then

```

```

!      Print solution

      Write (nout,*) 'Eigenvalues'
      Write (nout,99999) d(1:n)
      Flush (nout)

!      Standardize the eigenvectors so that first elements are non-negative.
      Do i = 1, n
        If (z(1,i)<0.0_nag_wp) Then
          z(1:n,i) = -z(1:n,i)
        End If
      End Do

!      ifail: behaviour on error exit
!      =0 for hard exit, =1 for quiet-soft, =-1 for noisy-soft
      ifail = 0
      Call x04caf('General',' ',n,n,z,ldz,'Eigenvectors',ifail)

!      Get the machine precision, EPS and compute the approximate
!      error bound for the computed eigenvalues. Note that for
!      the 2-norm, max( abs(D(i)) ) = norm(A), and since the
!      eigenvalues are returned in ascending order
!      max( abs(D(i)) ) = max( abs(D(1)), abs(D(n)) )

      eps = x02ajf()
      eerrbd = eps*max(abs(d(1)),abs(d(n)))

!      Call DDISNA (F08FLF) to estimate reciprocal condition
!      numbers for the eigenvectors
      Call ddisna('Eigenvectors',n,n,d,rcondz,info)

!      Compute the error estimates for the eigenvectors

      Do i = 1, n
        zerrbd(i) = eerrbd/rcondz(i)
      End Do

!      Print the approximate error bounds for the eigenvalues
!      and vectors

      Write (nout,*)
      Write (nout,*) 'Error estimate for the eigenvalues'
      Write (nout,99998) eerrbd
      Write (nout,*)
      Write (nout,*) 'Error estimates for the eigenvectors'
      Write (nout,99998) zerrbd(1:n)
      Else
        Write (nout,99997) 'Failure in DSTEV. INFO =', info
      End If

99999 Format (3X,(8F8.4))
99998 Format (4X,1P,6E11.1)
99997 Format (1X,A,I4)
      End Program f08jafa

```

## 10.2 Program Data

F08JAF Example Program Data

```

4                               :Value of N

1.0  4.0  9.0  16.0  :End of diagonal elements
1.0  2.0  3.0       :End of off-diagonal elements

```

### 10.3 Program Results

F08JAF Example Program Results

Eigenvalues

0.6476 3.5470 8.6578 17.1477

Eigenvectors

	1	2	3	4
1	0.9396	0.3388	0.0494	0.0034
2	-0.3311	0.8628	0.3781	0.0545
3	0.0853	-0.3648	0.8558	0.3568
4	-0.0167	0.0879	-0.3497	0.9326

Error estimate for the eigenvalues

1.9E-15

Error estimates for the eigenvectors

6.6E-16 6.6E-16 3.7E-16 2.2E-16

---